unittest
{
	assert( [ "2 60", "10 10", "100 100" ].solve == 110 );
	assert( [ "3 60", "10 10", "10 20", "10 30" ].solve == 60 );
	assert( [ "3 60", "30 10", "30 20", "30 30" ].solve == 50 );
	assert( [ "10 100", "15 23", "20 18", "13 17", "24 12", "18 29", "19 27", "23 21", "18 20", "27 15", "22 25" ].solve == 145 );
}



import std.algorithm;
import std.conv;
import std.format;
import std.range;
import std.stdio;

void main()
{
	stdin.byLineCopy.solve.writeln;
}

auto solve( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = 0L;
	auto t = 0L;
	input.front.formattedRead!"%s %s"( n, t );
	auto abs = new long[ 2 ][ n ];
	foreach( ref ab; abs )
	{
		input.popFront;
		ab = input.front.split.to!( long[] );
	}
	
	// dp[ i番目の料理 ][ 経過した時間 ][ ラストオーダー済フラグ ] = おいしさの合計
	auto dp = new long[][][]( 1 + n, t, 2 );
	foreach( i, ab; abs )
	{
		foreach( j; 0L .. t )
		{
			// 食べない
			dp[ i + 1 ][ j ][ 0 ] = max( dp[ i + 1 ][ j ][ 0 ], dp[ i ][ j ][ 0 ] );
			dp[ i + 1 ][ j ][ 1 ] = max( dp[ i + 1 ][ j ][ 1 ], dp[ i ][ j ][ 1 ] );
			
			// 食べる
			auto jj = j + ab[ 0 ];
			if( jj < t )
			{
				dp[ i + 1 ][ jj ][ 0 ] = max( dp[ i + 1 ][ jj ][ 0 ], dp[ i ][ j ][ 0 ] + ab[ 1 ] );
				dp[ i + 1 ][ jj ][ 1 ] = max( dp[ i + 1 ][ jj ][ 1 ], dp[ i ][ j ][ 1 ] + ab[ 1 ] );
			}
			
			// 最後に食べる
			dp[ i + 1 ][ j ][ 1 ] = max( dp[ i + 1 ][ j ][ 1 ], dp[ i ][ j ][ 0 ] + ab[ 1 ] );
		}
	}
	return dp[ $ - 1 ][ $ - 1 ].reduce!max;
}
